


Inno Setup Quick Guide

I/ Création du script

Au lancement d'un nouveau script apparaît la page de configuration de l'entête.
Cliquer sur suivant et remplir les champs :


- *Application name* : nom de l'application
- *Application version* : version de l'application
- *Application publisher* : Identité de la compagnie de distribution
- *Application website* : lien internet vers la source/ le site web



The screenshot shows the 'Inno Setup Script Wizard' window, specifically the 'Application Information' step. The window has a title bar with the text 'Inno Setup Script Wizard' and a close button. Below the title bar, the section is titled 'Application Information' with a subtitle 'Please specify some basic information about your application.' and a document icon. The main area contains four labeled text input fields: 'Application name:' with the value 'My Program', 'Application version:' with the value '1.5', 'Application publisher:' with the value 'My Company, Inc.', and 'Application website:' with the value 'http://www.example.com/'. At the bottom left, it says 'bold = required'. At the bottom right, there are three buttons: '< Back', 'Next >' (which is highlighted with a blue border), and 'Cancel'.


La page suivante concerne le chemin par default d'installation d'application (Program Files).

Il y a possibilité de changer le chemin et le nom du dossier d'installation.




The screenshot shows the 'Inno Setup Script Wizard' window, specifically the 'Application Folder' step. The title bar reads 'Inno Setup Script Wizard'. The main heading is 'Application Folder' with a subtitle 'Please specify folder information about your application.' and a folder icon with a star. The 'Application destination base folder:' section has a dropdown menu set to 'Program Files folder' and an empty text box below it. The 'Application folder name:' section has a text box containing 'My Program'. There are two checkboxes: 'Allow user to change the application folder' (checked) and 'The application doesn't need a folder' (unchecked). At the bottom, there is a legend 'bold = required' and three buttons: '< Back', 'Next >' (highlighted with a blue border), and 'Cancel'.

Puis définition du nom de programme principale et c'est à cet instant qu'on peut lier des fichiers et dossiers à l'installation. Les fichiers et dossiers liés seront compressés et intégrés au setup exécutable.



The screenshot shows the 'Inno Setup Script Wizard' window, specifically the 'Application Files' step. The title bar reads 'Inno Setup Script Wizard'. The main heading is 'Application Files' with a subtitle 'Please specify the files that are part of your application.' and a folder icon with a star. The 'Application main executable file:' section has a text box containing 'C:\Program Files (x86)\Inno Setup 5\Examples\MyProg.exe' and a 'Browse...' button. There are two checkboxes: 'Allow user to start the application after Setup has finished' (checked) and 'The application doesn't have a main executable file' (unchecked). The 'Other application files:' section has a large empty list box and four buttons: 'Add file(s)...', 'Add folder...', 'Edit...', and 'Remove'. At the bottom, there is a legend 'bold = required' and three buttons: '< Back', 'Next >' (highlighted with a blue border), and 'Cancel'.

On définit les options d'installation supplémentaires (création du raccourci bureau et de la barre des tâches)



The screenshot shows the 'Application Shortcuts' step of the Inno Setup Script Wizard. The window title is 'Inno Setup Script Wizard'. The subtitle is 'Application Shortcuts' with the instruction 'Please specify which shortcuts should be created for your application.' Below this, there are several options with checkboxes. The first option, 'Create a shortcut to the main executable in the common Start Menu Programs folder', is checked. Below it is a text box for 'Application Start Menu folder name:' containing the text 'My Program'. There are three more unchecked options: 'Allow user to change the Start Menu folder name', 'Allow user to disable Start Menu folder creation', and 'Create an Internet shortcut in the Start Menu folder'. Below these is another section 'Other shortcuts to the main executable:' with two unchecked options: 'Allow user to create a desktop shortcut' and 'Allow user to create a Quick Launch shortcut on older versions of Windows'. At the bottom left, it says 'bold = required'. At the bottom right, there are three buttons: '< Back', 'Next >' (which is highlighted with a blue border), and 'Cancel'.

Application Shortcuts
Please specify which shortcuts should be created for your application.

☒ Create a shortcut to the main executable in the common Start Menu Programs folder
Application Start Menu folder name:


☒ Allow user to change the Start Menu folder name
☐ Allow user to disable Start Menu folder creation
☐ Create an Internet shortcut in the Start Menu folder
☐ Create an Uninstall shortcut in the Start Menu folder

Other shortcuts to the main executable:
☒ Allow user to create a desktop shortcut
☐ Allow user to create a Quick Launch shortcut on older versions of Windows

bold = required

< Back **Next >** Cancel

On associe les fichiers de Licence, le guide d'installation et les informations de fermeture.



The screenshot shows the 'Application Documentation' step of the Inno Setup Script Wizard. The window title is 'Inno Setup Script Wizard'. The subtitle is 'Application Documentation' with the instruction 'Please specify which documentation files should be shown by Setup during installation.' Below this, there are three sections, each with a text box and a 'Browse...' button. The first section is 'License file:' with an empty text box and a 'Browse...' button. The second section is 'Information file shown before installation:' with an empty text box and a 'Browse...' button. The third section is 'Information file shown after installation:' with an empty text box and a 'Browse...' button. At the bottom right, there are three buttons: '< Back', 'Next >' (which is highlighted with a blue border), and 'Cancel'.

Application Documentation
Please specify which documentation files should be shown by Setup during installation.


License file:
 Browse...

Information file shown before installation:
 Browse...


Information file shown after installation:
 Browse...

< Back **Next >** Cancel

Le langage du setup file



Les options supplémentaires concernant le setup file en lui-même (dossier de destination, nom de l'exécutable, icône utilisée et éventuellement mot de passe de protection).



II/ Écriture du script

Le script utilisé est de format .iss. On y retrouve toutes les manipulations liées à l'exécution d'un fichier d'installation. Un fichier iss est divisé en plusieurs sections dont certaines optionnelles. J'ai dans mon exemple utilisé le script iss pour lier différents setup déjà existant au format .bat/python.

Le script est ici divisé en 9 sections :

- [1 \[Setup\]](#)

Regroupe les informations générées automatiquement par l'initialisation de script. On y retrouve les informations (modifiables) du setup file. Les balises `{#MyAppName}``{#MyAppVersion}`... correspondent aux champs remplis lors de l'exécution de l'initialisation. Il faut y substituer les réelles chaînes de caractère dans la logique du script.

```
[Setup]
; On retrouve les informations essentielles concernant l'application.
; Il s'agit des metadatas du setup et des paramètres d'initialisation.
; Un setup est authentifié par un identifiant unique, renouvelé à chaque nouvelle création.

; identifiant du setup
AppId={{C2DB07D6-2158-4971-B04D-A02F0F603D74}}
; nom de l'application
AppName={#MyAppName}
; version de l'application
AppVersion={#MyAppVersion}
; nom du distributeur
AppPublisher={#MyAppPublisher}
; adresse web du distributeur
AppPublisherURL={#MyAppURL}
; adresse web du service de support
AppSupportURL={#MyAppURL}
; adresse web du service d'update du logiciel
AppUpdatesURL={#MyAppURL}
; repertoire par default
DefaultDirName={pf}\{#MyAppName}
; nom de groupe par default
DefaultGroupName={#MyAppName}
; options
AllowNoIcons=yes
InfoBeforeFile=C:\Users\flim-users\Documents\Tests install\final\PyMoDAQ\README-install guide.txt
InfoAfterFile=C:\Users\flim-users\Documents\Tests install\final\PyMoDAQ\README.txt
OutputDir=C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\
OutputBaseFilename=Setup
Compression=lzma
SolidCompression=yes
```

Metadatas du setup file

- [2/ \[Languages\]](#)

Regroupe les informations de langue utilisée

```
[Languages]
; options de langues du setup
Name: "english"; MessagesFile: "compiler:Default.isl"
```

Options de langues

- [3/ \[Types\]](#)

Dans cette section on peut définir plusieurs profils d'installation.

Un type représente un chemin spécifique d'exécution, une balise qui va permettre de dresser l'exécution du setup. Un Type est constitué d'un nom et d'une description, éventuellement associés aux flags disponible

```
; Définir deux type d'installation : Full et Custom
[Types]
; installation complete sans selection des modules
Name: "full"; Description: "Full installation";
; installation personnalisée avec choix des modules à installer
Name: "custom"; Description: "Custom installation";
```

Définition des types d'installation

- [4/ \[Components\]](#)

Il s'agit des composantes représentées par les balises de type.

On y retrouve les profils full et custom définis dans la section type. Une composante est liée à un type et sert à router le chemin d'exécution le long du script. On lui associe un nom et une description.

```
; On lie aux type les composantes associées
[Components]
; Balise d'installation complète
Name: "full"; Description: "Full Installation"; Types: full
; Balise d'installation personnalisée
Name: "custom"; Description: "Custom Installation"; Types: custom
```

Définition des Composants de l'installation

- [5/ \[Tasks\]](#)

Ici on divise encore une fois le chemins d'exécution en nombre de tasks chemins différents. Les tasks ne seront visibles que dans le cadre d'une installation custom. Les tasks servent également de balises mais au niveau inférieur de la section Types. On peut associer des tasks à une composante, dans ce cas le task ne sera disponible que dans le profile d'installation décrit par la composante. On lui associe un nom, une description et un groupe.

On peut dresser une arborescence des tasks en utilisant le séparateur / pour définir le lien père/fils. Les tasks apparaîtront dans le setup file comme des checkbox à cocher.

Dans le cadre de tasks en arborescence l'installation du père engendre obligatoirement celle de tout ses fils.

```
; Definition des choix de modules selon la composante custom (apparaissent sous forme d'une checkbox dans le setup)
[Tasks]
; Balise d'installation du DAQ_Meteor, la balise est liée au Component custom
Name: "DAQ_Meteor"; Description: "DAQ Meteor module installer"; GroupDescription: "Modules:"; Components: custom;
; Balise d'installation du DAQ_Scan comme père des suivants (nécessitent les modules fils pour fonctionner)
; La balise est liée au Component custom
Name: "DAQ_Scan"; Description: "DAQ Scan module installer"; GroupDescription: "Modules:"; Components: custom;
; Balise d'installation du DAQ_Analysis comme fils de DAQ_Scan, la balise est liée au Component custom
Name: "DAQ_Scan\DAQ_Analysis"; Description: "DAQ Analysis module installer"; GroupDescription: "Modules:"; Components: custom;
; Balise d'installation du DAQ_Measurement comme fils de DAQ_Scan, la balise est liée au Component custom
Name: "DAQ_Scan\DAQ_Measurement"; Description: "DAQ Measurement module installer"; GroupDescription: "Modules:"; Components: custom;
; Balise d'installation du DAQ_Move comme fils de DAQ_Scan, la balise est liée au Component custom
Name: "DAQ_Scan\DAQ_Move"; Description: "DAQ Move module installer"; GroupDescription: "Modules:"; Components: custom;
; Balise d'installation du DAQ_Viewer comme fils de DAQ_Scan, la balise est liée au Component custom
Name: "DAQ_Scan\DAQ_Viewer"; Description: "DAQ Viewer module installer"; GroupDescription: "Modules:"; Components: custom;
```

Mise en place des Tasks représenté par des checkbox associées aux modules à installer

- [6/ \[Files\]](#)

Il s'agit de la section regroupant les fichiers/dossiers à copier lors de l'exécution du setup. En l'absence de balises **Components** ou **Tasks**, les fichiers sont copiés quelque soit le profil d'installation. Le champs **DestDir** contient le chemin relatif de destination de la copie. **{app}** est le nom du fichier root associé à l'installateur (le nom de l'application). Les flags '**recursesubdirs**' et '**createallsubdirs**' permettent la copie récursive des sous dossiers et sous fichiers. En associant à un fichier/dossier une balise **Components** on restreint la copie à ce seul profil d'installation. On peut réaliser une installation précise en liant également une balise tasks restreignant la copie au profil correspondant.

```
[Files]
;Installation des fichiers communs à tout les profils d'installation
; Il faut être vigilant à bien reconstruire la même arborescence d'application grace aux champs DestDir
; La specification C:\path\dossier\* associé aux flags recursebubdirs et createallsubdirs copie l'ensemble des
; sous-repertoires
Source: "C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\spyproject\*"; DestDir: "{app}\spyproject"; Flags: ignoreversion recursebubdirs createallsubdirs;
Source: "C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\_pycache\_"; DestDir: "{app}\_pycache\_"; Flags: ignoreversion recursebubdirs createallsubdirs;
Source: "C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\Documentation\*"; DestDir: "{app}\Documentation"; Flags: ignoreversion recursebubdirs createallsubdirs;
Source: "C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\Libraries_packages\*"; DestDir: "{app}\Libraries_packages"; Flags: ignoreversion recursebubdirs createallsubdirs;
Source: "C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\Manuals\*"; DestDir: "{app}\Manuals"; Flags: ignoreversion recursebubdirs createallsubdirs;
Source: "C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\QtDesigner_Ressources\*"; DestDir: "{app}\QtDesigner_Ressources"; Flags: ignoreversion recursebubdirs createallsubdirs;
; La specification C:\path\fichier copie uniquement le fichier correspondant
Source: "C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\__init__.py"; DestDir: "{app}";
Source: "C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\icon.ico"; DestDir: "{app}";
Source: "C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\LICENSE"; DestDir: "{app}";
Source: "C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\README.md"; DestDir: "{app}";
Source: "C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\README.txt"; DestDir: "{app}";
Source: "C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\README-install guide.txt"; DestDir: "{app}";
Source: "C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\welcome.html"; DestDir: "{app}";
```

Installation des fichiers communs aux différents profils

```
;Full installation bundle
; Ici le repertoire spécifié n'est copié que si le profil d'installation correspond à la Balise full
Source: "C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\*"; DestDir: "{app}"; Flags: ignoreversion recursebubdirs createallsubdirs; Components: full
```

Dans le cadre d'une installation complète (spécifiée par **Components : full), on copie l'ensemble du répertoire de l'application.**


```
;Custom installation files (depending on tasks section)
; De la même manière les répertoires ci-dessous ne sont copiés que si le profil d'installation correspond à la Balise custom
; et si de plus le tasks correspondant a été coché (dans ce cas la balise tasks sera active lors de l'exécution du setup).
Source: "C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\DAQ_Analysis\*"; DestDir: "{app}\DAQ_Analysis"; Flags: ignoreversion recursesubdirs createallsubdirs; Components: custom; Tasks: "DAQ_Scan\DAQ_Analysis";
Source: "C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\DAQ_Measurement\*"; DestDir: "{app}\DAQ_Measurement"; Flags: ignoreversion recursesubdirs createallsubdirs; Components: custom; Tasks: "DAQ_Scan\DAQ_Measurement";
Source: "C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\DAQ_Move\*"; DestDir: "{app}\DAQ_Move"; Flags: ignoreversion recursesubdirs createallsubdirs; Components: custom; Tasks: "DAQ_Scan\DAQ_Move";
Source: "C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\DAQ_Viewer\*"; DestDir: "{app}\DAQ_Viewer"; Flags: ignoreversion recursesubdirs createallsubdirs; Components: custom; Tasks: "DAQ_Scan\DAQ_Viewer";
Source: "C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\DAQ_Meteor\*"; DestDir: "{app}\DAQ_Meteor"; Flags: ignoreversion recursesubdirs createallsubdirs; Components: custom; Tasks: "DAQ_Meteor";
```

Installation des répertoires spécifiques aux différents profils d'une installation custom(spécifiée par **Components : custom).**

Les profils d'installation custom sont spécifiés par **Tasks**: DAQ_Meteor . Par ex un fichier ayant les balises **Components**: custom; **Tasks**: "DAQ_Scan\DAQ_Analysis"; ne sera copié que dans le cadre d'une installation custom et sous réserve que la checkbox DAQ_Analysis ai été coché.

- [7/ \[Code\]](#)

Il s'agit de la section dans laquelle on peut incorporer du code au setup, ici uniquement constitué d'une fonction appelant un script préliminaire à l'installation.

```
; Lancement du script pre-install verifiant la presence de python 3 sur le pc
[Code]
function InitializeSetup(): boolean;
var
    ResultCode: integer;
begin
    Exec(ExpandConstant('C:\Users\flim-users\Documents\Tests install\Final\PyMoDAQ\DAQ_Analysis\setup\pyAsk.bat'), '', '', SW_SHOW, ewWaitUntilTerminated, ResultCode)
    Result := True;
end;
```

Script de pré-installation vérifiant l'installation de python sur la machine hôte et invitant l'utilisateur à installer la dernière version de Python si besoin.

- [8/ \[Run\]](#)

Dans la fonction run on lie les scripts d'installation aux différents profils. Ici je n'utilise que des scripts .bat garantissant la bonne exécution sur n'importe quelle machine équipée de Windows. Les Run peuvent être associés aux profils d'installation via les mêmes balises component et Tasks.

Par ex un script ayant la balise *Components: full*; ne sera exécuté que dans le cadre d'une installation full indépendamment des tasks.

Le flag 64bit restreint l'installation aux systèmes 64bits.

[Run]

```
;Definition des scripts d'installation unitaire à associer aux différents profils d'installation  
; Les profils dépendent de la composante (full ou custom) et des tasks  
;(choix des modules dans le cadre d'une installation custom)
```

```
Filename: "{app}\DAQ_Analysis\setup\PyMoDAQ_dependencies-1.0.win-amd64.exe"; StatusMsg: "Building dependencies"; Flags: 64bit; Components: full;  
Filename: "{app}\DAQ_Analysis\setup\make_dep.bat"; StatusMsg: "Building dependencies"; Flags: 64bit; Components: full;  
Filename: "{app}\DAQ_Scan\setup\setup.bat"; StatusMsg: "Building module"; Flags: 64bit; Tasks: "DAQ_Scan"; Components: custom;  
Filename: "{app}\DAQ_Analysis\setup\setup.bat"; StatusMsg: "Building module"; Flags: 64bit; Tasks: "DAQ_Scan\DAQ_Analysis" ; Components: custom;  
Filename: "{app}\DAQ_Measurement\setup\setup.bat"; StatusMsg: "Building module"; Flags: 64bit; Tasks: "DAQ_Scan\DAQ_Measurement"; Components: custom;  
Filename: "{app}\DAQ_Meteor\setup\setup.bat"; StatusMsg: "Building module"; Flags: 64bit; Tasks: "DAQ_Meteor"; Components: custom;  
Filename: "{app}\DAQ_Move\setup\setup.bat"; StatusMsg: "Building module"; Flags: 64bit; Tasks: "DAQ_Scan\DAQ_Move"; Components: custom;  
Filename: "{app}\DAQ_Viewer\setup\setup.bat"; StatusMsg: "Building module"; Tasks: "DAQ_Scan\DAQ_Viewer"; Components: custom;
```

Les profils d'exécutions sont encore une fois spécifiés par les balises *Components* (full or custom) et *Tasks*

- [9/ \[Icons\]](#)

Dans cette section on lie les icônes et les raccourcis aux fichiers à exécuter.

La création de raccourcis est également soumise aux différents profils d'installation via les balises Components et Tasks

```
[Icons]
; Lier les icones aux executables
; Les icones présentes ici seront installées dans le menu demarrer de windows
Name: "{group}\DAQ Analysis"; Filename: "{app}\DAQ_Analysis\DAQ Analysis.exe"; IconFilename: "{app}/icon.ico";
Name: "{group}\DAQ Measurement"; Filename: "{app}\DAQ_Measurement\DAQ Measurement.exe"; IconFilename: "{app}/icon.ico";
Name: "{group}\DAQ Metheor"; Filename: "{app}\DAQ_Metheor\DAQ Metheor.exe"; IconFilename: "{app}/icon.ico";
Name: "{group}\DAQ Move"; Filename: "{app}\DAQ_Move\DAQ Move.exe"; IconFilename: "{app}/icon.ico";
Name: "{group}\DAQ Scan"; Filename: "{app}\DAQ_Scan\DAQScan.bat"; IconFilename: "{app}/icon.ico";
Name: "{group}\DAQ Viewer"; Filename: "{app}\DAQ_Viewer\DAQ Viewer.exe"; IconFilename: "{app}/icon.ico";
Name: "{group}\PyMoDAQ Documentation"; Filename: "{app}\Documentation\_build\html\index.html";
Name: "{group}\uninstall PyMoDAQ"; Filename: "{app}\unins000.exe";
Name: "{group}\{cm:ProgramOnTheWeb,{#MyAppName}}"; Filename: "{#MyAppURL}";
```

Création du menu PyMoDAQ dans le menu démarrer/Programmes de Windows

```
;Creation des raccourcis selon le profil d'installation
```

```
;Full installation shortcuts profile
```

```
; Creation des raccourcis bureau d'une installation complète
```

```
Name: "{userdesktop}\DAQ Scan"; Filename: "{app}\DAQ_Scan\DAQScan.bat"; IconFilename: "{app}/icon.ico"; Components: full;
Name: "{userdesktop}\DAQ Metheor"; Filename: "{app}\DAQ_Metheor\DAQ Metheor.exe"; IconFilename: "{app}/icon.ico"; Components: full;
Name: "{userdesktop}\DAQ Analysis"; Filename: "{app}\DAQ_Analysis\DAQ Analysis.exe"; IconFilename: "{app}/icon.ico"; Components: full;
Name: "{userdesktop}\DAQ Measurement"; Filename: "{app}\DAQ_Measurement\DAQ Measurement.exe"; IconFilename: "{app}/icon.ico"; Components: full;
Name: "{userdesktop}\DAQ Move"; Filename: "{app}\DAQ_Move\DAQ Move.exe"; IconFilename: "{app}/icon.ico"; Components: full;
Name: "{userdesktop}\DAQ Viewer"; Filename: "{app}\DAQ_Viewer\DAQ Viewer.exe"; IconFilename: "{app}/icon.ico"; Components: full;
```

Création des raccourcis bureau dans le cadre d'une installation complète

```
;Custom installation shortcuts profile
; Création des raccourcis bureaux correspondant à une installation custom
; Sous réserve d'avoir été coché par l'utilisateur.
Name: "{userdesktop}\DAQ Metheor"; Filename: "{app}\DAQ_Metheor\DAQ Metheor.exe"; IconFilename: "{app}/icon.ico";Tasks: "DAQ_Metheor"; Components: custom;
Name: "{userdesktop}\DAQ Scan"; Filename: "{app}\DAQ_Scan\DAQScan.bat"; IconFilename: "{app}/icon.ico"; Tasks: "DAQ_Scan"; Components: custom;
Name: "{userdesktop}\DAQ Analysis"; Filename: "{app}\DAQ_Analysis\DAQ Analysis.exe"; IconFilename: "{app}/icon.ico";Tasks: "DAQ_Scan\DAQ_Analysis"; Components: custom;
Name: "{userdesktop}\DAQ Measurement"; Filename: "{app}\DAQ_Measurement\DAQ Measurement.exe"; IconFilename: "{app}/icon.ico";Tasks: "DAQ_Scan\DAQ_Measurement"; Components: custom;
Name: "{userdesktop}\DAQ Move"; Filename: "{app}\DAQ_Move\DAQ Move.exe"; IconFilename: "{app}/icon.ico";Tasks: "DAQ_Scan\DAQ_Move"; Components: custom;
Name: "{userdesktop}\DAQ Viewer"; Filename: "{app}\DAQ_Viewer\DAQ Viewer.exe"; IconFilename: "{app}/icon.ico";Tasks: "DAQ_Scan\DAQ_Viewer"; Components: custom;


;Rajout du lien vers la documentation commun
Name: "{userdesktop}\Documentation"; Filename: "{app}\Documentation\_build\html\index.html";
```

Création des raccourcis bureau dans le cadre d'une installation custom.

Par exemple un raccourci ayant comme balises *Tasks: "DAQ_Metheor"; Components: custom;* ne sera créé sur le bureau que dans le cadre d'une installation custom sous réserve que la checkbox DAQ_Metheor ai été cochée.

III/ Exécution et compilation

A chaque moment on peut vérifier la bonne exécution du script en l'interprétant. Pour cela il suffit de cliquer sur la flèche verte pour démarrer le script.



La compilation est accessible via le menu build/compile, le fichier setup.exe correspondant sera créé dans le répertoire spécifié en section [Setup] champs OutputDir.

Une fois compilé, le setup comprend tout les fichiers associés compressés, les scripts d'installation nécessaires ainsi que le script iss d'exécution.

Sous forme d'un exécutable Windows, il ne suffit plus qu'à démarrer l'installation.